

Scott Baker



Atari
Atari BASIC
48K

5/1/84

Superlearner

I was very intrigued by "The Learner" program featured in the beginner's page of the September issue of COMPUTE! Inspired by the original 23-line program I created a 266-line menu-driven program called Superlearner.

Superlearner's files are nearly as large as the program itself and it is therefore able to store up to 30 different facts on each of ten different subjects. Up to 20 of these facts may explain what a particular subject is. For example, You may teach Superlearner that a **Personal Computer** (subject) is a **device containing a microprocessor** (fact). However, since Superlearner is also able to learn up to ten different facts about what a subject is not, it is possible as well to teach Superlearner that a **Personal Computer** is not **Aunt Gertie**. Superlearner accepts silly facts with equal aplomb. I'll describe exactly how to do all this, and more, later.

Writing the program taught me a lot about what intelligence--both the real and artificial kind--is and is not. The first thing that became apparent as Superlearner became more capable of storing and retrieving large amounts of information was its total inability to relate one subject to another based on shared facts. That's the inspiration behind the essay-writing routine on lines 550-998. Frankly, the rest of the program is partly designed to prepare Superlearner to write a quality essay from this routine, although all the other menu functions stand alone when called upon.

Running the program

After you have saved the program and ran it, Superlearner

will greet you with a message reminding you of its capabilities. The first question you will then be asked is whether you will be using a printer. Incidentally, your answers to all prompts will automatically appear in lower-case. Lower-case is Superlearner's default mode so that any information you later teach Superlearner will blend into its own lower-case sentence fragments when it's asked to compose answers and essays. Before you answer yes to the first question, you must turn your printer on or an ERROR-138 will result when Superlearner tries to open a channel to a "non-existent" printer. Furthermore, if you answer "no" here by simply pressing RETURN you can't use a printer during the rest of the program.

The next question Superlearner will ask you is whether you want a demonstration of its learning abilities. A "y" here will cause the program to branch to the routine on lines 1200-1340. In this routine, Superlearner will read in eight positive facts and four negative facts about the subject of a panda and the same amount about a dog. The information absorbed in this routine fills less than 8% of Superlearner's information storage capacity, yet it is still enough to produce some interesting essays and to experiment with all the other features of the program.

The blank slate approach

After using Superlearner a few times, you will probably not want to use the demo information. To keep this information from being read in, simply press RETURN in response to the "Do you want a demonstration..." question. Superlearner will then ask you if it should retrieve information previously stored on tape or

disk. A "y" here will branch the program to the tape/disk reading routine on lines 1050-1110. You should then specify whether the information to be retrieved is on tape or disk. If you respond with "d:" Superlearner will ask for the filename as well. (Caps are temporarily POKEd back for your answer.)

After the information has been read in, you will be presented with the menu of options. More on that later. Attention tape-users: it takes nearly 15 minutes to store or retrieve the roughly 13K of information on tape, even with the screen turned off and other time-saving devices employed!

If you haven't been skipping ahead, you probably haven't saved any information yet so I'll return to this option later.

By answering with a RETURN to the question regarding tape/disk retrieval, you cause Superlearner to ask you to supply the first subject to be discussed. This prompt is easy to answer. The subject you choose--including a leading article if necessary--must be no more than 20 characters long or Superlearner will flash the message "That's too much for me to remember!" and make a you-shouldn't-have-done-that sound. You will also get this reaction if you ever try to teach Superlearner a fact longer than 40 characters, exceed ten subjects, or try to make Superlearner remember more than 20 positive facts about a subject or ten negative facts about a subject. In every case where the overload routine is invoked you will be given another chance to answer again with no harm done.

The options menu

After you have either 1) chosen your own first subject 2)

allowed Superlearner to retrieve information on tape or disk or
 3) allowed Superlearner to read in its own demo information, you
 will now be presented with the menu. Typing in the following
 characters--in lower case only--and pressing RETURN--will execute
 the associated options.

q -- Question Superlearner: To ask Superlearner a question
 about the subject currently mentioned in the menu, complete your
 question after the prompt "Is it..." and Superlearner will check
 for an **exact** match in both its positive and negative files on the
 subject. If Superlearner doesn't know the answer it will ask you
 to teach it. By the way, if you enter the question option by
 mistake, simply press RETURN at the "Is it..." prompt and you
 will be returned to the menu. Do not press RETURN at the "Please
 teach me" prompt or you will store a blank space as a fact.

a -- in this option you can ask Superlearner to match a
 specific question to all subjects. For example, you could ask
 Superlearner "Is anything a device with a microprocessor" and it
 might respond "yes...a computer is device with a microprocessor"
 followed by, "In fact, a tree is not a device with a
 microprocessor" provided, of course, that you had previously
 taught it these facts. In this option, Superlearner will not ask
 you to teach it information it doesn't already know.

l -- List all facts about a subject: Typing an "l" will
 produce a list of all a subject is, followed by the option to
 print this list to the printer. After you answer this prompt,
 Superlearner will list all a subject is not. You also have the
 option of printing out this list.

e -- List all facts about all subjects: When you use these

fact listings, information on all subjects will be devulged with a prompt between each listing to allow you to print it to your printer. This option is the best way to get a hardcopy of all information on all subjects.

o -- Return to an old subject: After entering "o" answer the prompt with the subject you would like to discuss. If you ask Superlearner to talk about a subject it doesn't know, it will list all the subjects it does know and ask you to enter your choice again. You must select one of these choices before Superlearner will return to the menu.

n -- Begin a new subject: Provided you haven't already filled Superlearner up with ten subjects, entering "n" will allow you to enter a new subject to be discussed.

f -- Forget a subject or a fact: Unlike people, Superlearner can unlearn a single fact or even an entire subject and all the facts about it. Again, if you entered this option by mistake, simply press RETURN in response to the first question "Do you want me to forget a subject or just a fact..." and you will be returned to the menu with nothing lost. If you answer this question with an "s" Superlearner will ask you "which subject?". If you don't enter a subject Superlearner knows about, it will ask you to reenter your choice. When you have answered correctly, Superlearner will purge itself of the subject and all attendant information.

To forget just a fact enter an "f" for the first question. Superlearner will first ask you for the subject in which the fact occurs and then for the specific fact. Note that both positive

and negative fact files are checked for the fact, though if the fact occurs in both (e.g. a logical contradiction like **a frog is green** and **a frog is not green**) only the fact in the positive file will be erased. Use the forget option again to erase the fact from the negative fact file.

Also, be aware that although Superlearner can forget the tenth (last) subject and all the facts about it, it can't forget the tenth (last) negative fact or the twentieth (last) positive fact in a subject alone. This is because facts for all subjects are stored in the strings F\$ and N\$ and when you ask Superlearner to forget a single fact--instead of an entire subject--that fact is actually "covered up" by the fact in the next higher position in the string. In the case of the last fact in the substring allocated for the facts of a particular subject, the next fact would probably be the first fact of the substring allocated for the next subject! Got that?!

In any case, to have Superlearner forget say, the twentieth positive fact, you must first make Superlearner forget the nineteenth fact--or any fact before the twentieth--and then use the forget option again to make Superlearner forget the new nineteenth fact, which was formally the twentieth fact. Finally, you will probably want to reteach the original nineteenth fact back again.

If you ask Superlearner to forget a fact it never knew to begin with in the subject you selected, you will be informed of Superlearner's confusion and returned to the menu. This would probably be a good time to list all the information about the subject in question so you can check your spelling, case, etc.

Finally, if everything has gone right, you will be informed that Superlearner has forgotten whether or not the _____ (subject) is _____ (fact) and returned to the menu.

RETURN -- Teach Superlearner Pressing RETURN alone will cause Superlearner to prompt you to teach it a new fact. Answers to this prompt should be no more than 40 characters long or the overload routine will be invoked. To include a negative fact about a subject simply write the fact as if were positive but type a minus sign after the last character of the fact. (Of course, this means that negative facts can really be only up to 39 characters long plus the minus sign). Positive facts are taught without anything following the last character. Superlearner will acknowledge whether it has learned that a subject is some fact or is not some fact.

s -- Save all information The saving routine on lines 1000-1045 is pretty straight-forward to use although the code involves some interesting and advanced concepts of data storage. After you have answered the tape or disk question and entered the filename if neccessary, the screen will go blank to speed up execution time. When all information is stored, the menu will reappear. For cassette storage, be prepared to use ten minutes of high-quality tape -- no matter how much actual information is in the files -- and I would strongly advise backup saves as well.

w -- Write an essay: The write option is where Superlearner really gets to express itself. When the program branches to line 550 you will be asked if you want Superlearner to write about everything it knows or just about a single subject from its

files. I find the first option more interesting and surprising.

The next option you are given is to enter the title. The title must be no more than 40 characters.

Third, Superlearner asks if you want a printed copy as well as an essay on the screen.

Fourth, for those privileged few who have the Atari 1450XLD you may now elect to have Superlearner speak as well as write! Sorry, 800 owners, typing a "y" here will only return you to the menu, as will any other error you commit in answering the questions above.

Finally, Superlearner asks you how many pages it should write. Decimal notation and/or whole numbers are fine.

Here's a general breakdown of the main writing loop:

582-585 -- These lines select a subject at random provided you haven't previously chosen a single-subject essay. Also, Superlearner is likely to write more about the subjects it knows the most about, due to the higher probability of a high-information subject being selected in this routine.

590 and 591 -- Superlearner stresses the positive since the program branches two thirds of the time to line 590 and only one third of the time to line 591. Line 590 gets positive facts about a subject at random and line 591 gets negative facts about a subject at random.

600 and 650 cause a random branch to the appropriate lines where the subject and fact are spliced into sentence fragments in the appropriate places.

Note that throughout the essay, the ? #J; command is used so that text may be printed to the screen when J=1, to the

printer when J=2 and, in the 1450XLD, even to the speech synthesizer when J=4. This saves having to write nearly identical PRINT statements three times although it also results in a compromise fit between an 80-column printer and the 40-column screen.

900-998 -- Two-thirds of the time when Superlearner is writing about more than one subject, it will drop down to line 621 and then to the subroutine on lines 900-998 where Superlearner compares subjects which share the same facts. This routine is probably the most creative thing Superlearner does. It works best if you have previously taught Superlearner subjects that share several facts in common--even if a particular fact is what the subject is in one case and what it is not in another. For example, Superlearner might point out that "...although your dog Fido is a house pet, Clarence the lion is not."

780-820 -- This routine allows Superlearner to smooth the written transition from one subject to another.

691-735 -- When Superlearner is all done it will write a wrapup. What it says depends on how much you have taught it and whether it has written a single-subject essay or an all-subject essay. After this goodbye, you will be returned to the menu once again.

Type the program exactly as shown and your Atari will be transformed into an eager seeker of knowledge!


```

21 REM ****Superlearner by Scott Baker
10 B=20:V=40:DIM F$(10*B*V),FL(9,19),N$(10*10*V),FL1(9,9),SUB$(200),TSUB$(21),SU
BL(9),F(9),N(9),FACT$(41),QUE$(41)
15 S=0:T=1:DIM D$(8),X$(T),DEC$(T),C$(11):FOR X=S TO 9:F(X)=S:N(X)=S:SUBL(X)=S:N
EXT X:L=123:M=130:A=619:U=2
20 O=400:G=978:CT=-T:F$="" :F$(8000)=F$:F$(U)=F$:N$="" :N$(4000)=N$:N$(U)=N$:SUB
$="" :SUB$(200)=SUB$:SUB$(U)=SUB$
30 OPEN #T,8,S,"S:":? :? " Hello, I am Superlearner.":? " You can teach me up
to ten":? "different subjects."
40 Z=665:D=129:POKE 702,S:? " In each subject I can learn up to 20 facts abou
t what a subject is and"
50 ? "10 facts about what a subject is not.":? :? " Please enter today's date":
? "(mm/dd/yy)":;INPUT D$
60 ? :? " Will you be using a printer":? "(y/n)":;INPUT X$:IF X$="y" THEN OPEN
#U,8,S,"P:"
65 ? :? " Do you want a demonstration of my":? "learning abilities":? "(y/n)":;
INPUT DEC$
70 IF DEC$="y" THEN ? "):GOSUB 1200:GOTO M
80 ? :? " Should I retrieve information on":? "tape/disk":? "(y/n)":;INPUT DEC$
:IF DEC$="y" THEN 1050
90 ? :? " What is the general class of":? "subjects":;INPUT C$:IF LEN(C$)>10 TH
EN GOSUB 0:GOTO 90
110 ? :? "The ";CT+U;" subject we will discuss is":IF CT=9 THEN GOSUB 0:GOTO L
120 INPUT TSUB$:IF LEN(TSUB$)>B THEN GOSUB 0:GOTO 110
121 CT=CT+T:C=CT:SUBL(C)=LEN(TSUB$):SUB$(C*B+T,C*B+B)=TSUB$:GOTO M
123 ? :? " What subject would you like to talk about":;INPUT TSUB$:IF LEN(TSUB$
)>B THEN GOSUB 0:GOTO L
125 FOR C=S TO CT:IF SUB$(C*B+T,C*B+SUBL(C))=TSUB$ THEN ? " What would you like
to know about":? TSUB$;"?":GOTO M
126 NEXT C:GOSUB 127:GOTO L
127 ? :? "Sorry, I only know about...":FOR C=S TO CT:? SUB$(C*B+T,C*B+SUBL(C));"
, ";:NEXT C:? :RETURN
129 C=Y:TSUB$=SUB$(C*B+T,C*B+SUBL(C))
130 TRAP 40000:? :? "TYPE":? :? "q To question me about":? " ";TSUB$:CK=S
131 ? "a To question me about all ";C$:? "l To list everything about":? " ";TSU
B$
132 ? "e To list everything about ";C$:? "o To talk about an old subject":? "n T
o talk about a new subject"
133 ? "w To have me write an essay":? "f To forget a fact or a subject":? "s To
save information on tape/disk"
140 ? :? " Press RETURN alone to teach me":? "something new about ";TSUB$:INPUT
DEC$
158 IF DEC$="s" THEN 1000
159 IF DEC$="q" THEN 260
160 IF DEC$="l" THEN ? "Shall I list everything about":? TSUB$;" on the printer
(y/n)":;INPUT X$:POKE 752,T:? "~"
161 IF DEC$="l" THEN J=1:GOSUB 320:POKE 752,S:GOTO M
162 IF DEC$="n" THEN 110
163 IF DEC$="o" THEN GOTO L
164 IF DEC$="a" THEN Y=C:GOTO 500
165 IF DEC$="f" THEN ? "):GOTO 210
166 IF DEC$="e" THEN ? "List everything about ";C$:? "on the printer (y/n)":;INP
UT X$:POKE 752,T:? "~"
167 IF DEC$="e" THEN J=1:Y=C:GOTO 529
168 IF DEC$="w" AND F(C)>0 THEN Y=C:E=S:? "):GOTO 550

```



```

169 REM ****TEACH NEW INFORMATION
170 ? "What should I know"? :? " (Type a - after the fact if it's":? " what ";T
SUB$;" is not)"
180 TRAP M: ? "That it's...";:INPUT FACT$:IF LEN(FACT$)>V THEN GOSUB 0:GOTO M
190 IF FACT$(LEN(FACT$)<>"-") THEN GOSUB 409:GOTO 200
195 GOSUB 414: ? :? "Thanks...":? "I have learned that":? TSUB$;" is not ":? FACT
$(T,LEN(FACT$)-T);".":GOTO M
200 ? :? "Thanks...":? "I have learned that":? TSUB$;" is ":? FACT$;".":GOTO M
209 REM ****FORGET SUBJECT OR FACT
210 ? " Do you want me to forget a subject or just a fact (s=subject, f=fact)"
;:INPUT X$:IF X$="" THEN GOTO M
215 ? " Which subject";:INPUT TSUB$:FOR C=S TO CT:IF SUB$(C*B+T,C*B+SUBL(C))=TS
UB$ THEN 225
220 NEXT C:GOSUB 127:GOTO 215
225 IF X$="f" THEN ? " Which fact. That ";TSUB$:? "is/is not...";:INPUT QUE$:GO
SUB 420:CK=CK+T
226 IF X$="f" AND (I=9 AND CK=3) OR I=19 THEN ? "I'm unable to forget the last f
act about ";TSUB$:GOTO M
227 IF X$="f" THEN ON CK GOTO 250,251,253
230 TRAP 243:FOR X=C TO CT:SUB$(X*B+T,X*B+B)=SUB$((X+T)*B+T,(X+T)*B+B):SUBL(X)=S
UBL(X+T):NEXT X
233 FOR X=C TO CT:FOR I=S TO 18:FL(X,I)=FL(X+T,I):NEXT I:NEXT X
235 FOR X=C TO CT:F$(X*800+T,X*800+800)=F$((X+T)*800+T,(X+T)*800+800):F(X)=F(X+T
):NEXT X
238 FOR X=C TO CT:FOR I=S TO 8:FL1(X,I)=FL1(X+T,I):NEXT I:NEXT X
240 FOR X=C TO CT:N$(X*400+T,X*400+400)=N$((X+T)*400+T,(X+1)*400+400):N(X)=N(X+T
):NEXT X
243 CT=CT-T: ? " I have forgotten all about":? TSUB$;".":IF CT=-T THEN 110
244 IF C<9 THEN GOTO L
245 FOR X=T TO B:SUB$(C*B+X,C*B+X)=" ":NEXT X:SUBL(C)=S:F(C)=S:N(C)=S:F$(C*800+T
)=" ":F$(C*800+800)=F$(C*800+T)
247 F$(C*800+U)=F$(C*800+T):N$(C*400+T)=" ":N$(C*400+400)=N$(C*400+T):N$(C*400+U
)=N$(C*400+T):GOTO L
250 ? " I never knew whether ";TSUB$:? "was ";QUE$:? "or not in the first place
!":GOTO M
251 FOR X=I TO F(C)-U:F$(C*800+(X*V)+T,C*800+(X*V)+V)=F$(C*800+(X+T)*V+T,C*800+(
X+T)*V+V)
252 FL(C,X)=FL(C,X+T):NEXT X:F(C)=F(C)-T:GOTO 257
253 FOR X=I TO N(C)-U:N$(C*400+(X*V)+T,C*400+(X*V)+V)=N$(C*400+(X+T)*V+T,C*400+(
X+T)*V+V)
254 FL1(C,X)=FL1(C,X+T):NEXT X:N(C)=N(C)-T
257 ? :? " I have forgotten whether or not":? TSUB$;" is ";QUE$;".":GOTO M
259 REM ****CHECK KNOWLEDGE OF SUBJECT
260 TRAP M: ? :? " Ask me about ";TSUB$:? "Is it...";:INPUT QUE$:GOSUB 420:ON CK
GOTO 270,271
265 ? " You haven't taught me whether":? TSUB$;" is":? QUE$:? "Is it ";QUE$;" (
y/n)"::INPUT X$:GOTO 290
270 ? "Yes," :? TSUB$:? " is ";QUE$;".":GOTO M
271 ? "No," :? TSUB$:? " is not ";QUE$;".":GOTO M
290 IF X$="y" THEN FACT$=QUE$:GOSUB 409: ? "Thanks...":? "I have learned that":?
TSUB$;" is":? QUE$;".":GOTO 300
295 FACT$=QUE$:FACT$(LEN(FACT$)+T)="-":GOSUB 414
296 ? "Thanks...":? "I have learned that":? TSUB$;" is not":? QUE$;".
300 ? " You learn something new every day.":GOTO M

```



```

319 REM ****LIST ALL A SUBJECT IS
320 ? #J;"A listing of everything about ";TSUB$,D$:IF J=1 AND X$="Y" THEN J=2:GO
TO 320
330 J=1:QUE$="Press RETURN to continue...":IF F(C)=S THEN 350
331 ? #J: ? #J;TSUB$;" is...":FOR I=S TO F(C)-T: ? #J;F$(C*800+(I*V)+T,C*800+(I*V)
+FL(C,I));", ";:NEXT I: ? #J
340 IF X$="Y" AND J=1 THEN J=2:GOTO 331
349 REM ****LIST ALL A SUBJECT IS NOT
350 J=1:IF N(C)=S THEN 365
351 ? #J: ? #J;TSUB$;" is not...":FOR I=S TO N(C)-T: ? #J;N$(C*400+(I*V)+T,C*400+(
I*V)+FL1(C,I));", ";:NEXT I: ? #J
360 IF X$="Y" AND J=1 THEN J=2:GOTO 351
365 ? :IF X$<>"Y" THEN ? QUE$;:INPUT DEC$
366 RETURN
399 REM ****INFORMATION OVERLOAD
400 ? "":POKE 752,T:POSITION U,12: ? "That's too much for me to remember!"
401 SOUND S,150,10,10:FOR X=S TO 200:NEXT X:SOUND S,S,S,S:POKE 752,S: ? "":RETUR
N
408 REM ****ADD A POSITIVE FACT TO A SUBJECT
409 IF F(C)=B THEN FACT$=" ":GOSUB 0:GOTO M
410 FL(C,F(C))=LEN(FACT$):F$(C*800+F(C)*V+T,C*800+F(C)*V+V)=FACT$:F(C)=F(C)+T:RE
TURN
413 REM ****ADD A NEGATIVE FACT TO A SUBJECT
414 IF N(C)=10 THEN FACT$=" ":GOSUB 0:GOTO M
415 FL1(C,N(C))=LEN(FACT$)-T:N$(C*400+N(C)*V+T,C*400+N(C)*V+V)=FACT$(T,FL1(C,N(C)
)):N(C)=N(C)+T:RETURN
419 REM ****CHECK FOR POSITIVE FACTS ABOUT SUBJECT
420 IF F(C)=S THEN 430
421 FOR I=S TO F(C)-T:IF QUE$=F$(C*800+(I*V)+T,C*800+(I*V)+FL(C,I)) THEN CK=T:RE
TURN
425 NEXT I
429 REM ****CHECK FOR NEGATIVE FACTS ABOUT SUBJECT
430 IF N(C)=S THEN RETURN
431 FOR I=S TO N(C)-T:IF QUE$=N$(C*400+(I*V)+T,C*400+(I*V)+FL1(C,I)) THEN CK=U:R
ETURN
435 NEXT I:CK=S:RETURN
499 REM ****CHECK KNOWLEDGE OF ALL SUBJECTS
500 ? : ? " Ask me about any subject I know.": ? "Is anything..."::INPUT QUE$:IF
LEN(QUE$)>V THEN GOSUB 0:GOTO 500
505 X=S:FOR C=S TO CT:CK=S:GOSUB 420:IF CK>S THEN X=T:ON CK GOTO 515+INT(RND(S)*
5),517+INT(RND(S)*3)
508 NEXT C:IF X=S THEN ? " I haven't learned if anything": ? "is ";QUE$;" or not
!": ? " Please teach me.":GOTO D
510 ? : ? "That's all I've learned so far.": ? "Press RETURN to get menu"::INPUT X
$:GOTO D
515 ? "Yes,":GOTO 524
516 ? "Affirmative,":GOTO 524
517 ? "As a matter of fact,":GOTO 523+CK
518 ? "In fact,":GOTO 523+CK
519 ? "Well, I've been taught that":GOTO 523+CK
524 ? SUB$(C*B+T,C*B+SUBL(C));" is ";QUE$;".":GOTO 508
525 ? SUB$(C*B+T,C*B+SUBL(C));" is not ";QUE$;".":GOTO 508
528 REM ****LIST WHAT EVERYTHING IS
529 ? #J;"A listing of everything about ";C$,D$:IF J=1 AND X$="Y" THEN J=2:GOTO
529
530 FOR C=S TO CT:IF F(C)>S THEN TSUB$=SUB$(C*B+T,C*B+SUBL(C)):GOSUB 330
531 IF N(C)>S THEN TSUB$=SUB$(C*B+T,C*B+SUBL(C))
540 NEXT C:POKE 752,S:GOTO D

```



```

549 REM ****REM SUPERLEARNER'S ESSAY
550 RESTORE 557:TRAP D:? " Should I write about everything I know (y/n)";TSU
B$=C$:INPUT X$:IF X$="Y" THEN 557
551 ? :? " What subject should I write":? "about";:INPUT TSUB$:IF LEN(TSUB$)>B
THEN GOSUB 0:GOTO 551
552 X=C:FOR C=S TO CT:IF SUB$(C*B+T,C*B+SUBL(C))=TSUB$ THEN E=T:GOTO 554
553 NEXT C:GOSUB 127:C=X:GOTO 551
554 FOR X=S TO INT(RND(S)*3):READ QUE$:NEXT X:QUE$(LEN(QUE$)+T)=" on":GOTO 560
557 FOR X=T TO INT(RND(S)*CT):READ QUE$:NEXT X:QUE$(LEN(QUE$)+T)=" about":DATA T
oday's Lesson,An Essay,Opinions
558 DATA A Discussion,A Superlearner Exclusive,A Critical Essay,On The Many Thin
gs I Learned,The Nature of Things
560 ? " I always write on the screen, but should I write on the printer too":
? "(y/n)";:R=S:CK=S:INPUT DEC$
570 ? " About how many pages should":? "I write";:INPUT NS:NS=NS*B:?
573 ? " Press SELECT to stop my writing,"?:? "release to continue.":FOR X=S TO 2
50:NEXT X:TRAP 40000:J=T
575 POKE 752,T:? #J;QUE$;" ";TSUB$,D$:? #J:? #J;"I am Superlearner. This is what
I learned today...":? #J
576 IF DEC$="Y" AND J=T THEN J=U:GOTO 575
580 X1=S:W=-U:Q=S:FOR X=T TO NS:J=T
581 IF PEEK(53279)=5 THEN 581
582 X1=X1-T:IF X1>S OR E=T OR NS<X+3 THEN 586
583 X1=S:C=S:FOR H=S TO CT:R=INT(RND(S)*SQR(F(H)+N(H))):IF R>X1 THEN X1=R:C=H
585 NEXT H:QUE$=TSUB$:TSUB$=SUB$(C*B+T,C*B+SUBL(C)):W=W+U:IF QUE$<>TSUB$ THEN GO
SUB 780
586 ON INT(RND(S)*3)+T GOTO 590,590,591
590 IF F(C)>S THEN I=INT(RND(S)*F(C)):FACT$=F$(C*800+(I*V)+T,C*800+(I*V)+FL(C,I)
):GOTO 600
591 IF N(C)>S THEN I=INT(RND(S)*N(C)):FACT$=N$(C*400+(I*V)+T,C*400+(I*V)+FL1(C,I)
):GOTO 650
595 X=X-T:GOTO 690
600 IF X>T THEN R=INT(RND(S)*8):IF Q<>R THEN Q=R:RESTORE 740+Q:FOR R=S TO RND(S)
*3:READ QUE$:NEXT R:GOTO 604+Q
601 IF X>T THEN 600
602 RESTORE 740:READ QUE$
604 ? #J;"One of the ";QUE$;" is that":? #J;TSUB$;" is ";FACT$;".":GOTO A
605 ? #J;"Another ";QUE$;" you taught me is that ";TSUB$;" is ";FACT$;".":GOTO A
606 ? #J;"I am also ";QUE$;" that ";TSUB$;" is ";FACT$;".":GOTO A
607 ? #J;"It's ";QUE$;" to know that":? #J;TSUB$;" is ";FACT$;".":GOTO A
608 ? #J;"I ";QUE$;" what I already know about":? #J;TSUB$;" Would you teach me
more?":GOTO A
609 ? #J;"I must ";QUE$;" about ";TSUB$;".":? #J;"I didn't know it was ";FACT$;"
!":GOTO A
610 ? #J;"The fact that ";TSUB$;" is ";FACT$:? #J;"has ";QUE$;" that I am just b
eginning to realize.":GOTO A
611 ? #J;"I especially want to":? #J;QUE$;" that being ";FACT$:? #J;"is typical
of ";TSUB$;".
612 ? #J;"I'm learning so much about ";TSUB$;".
619 IF DEC$="Y" AND J=T THEN J=U:GOTO 604+Q
621 IF E=S AND CT>S THEN GOSUB 900
624 GOTO 690
650 IF X>T THEN R=INT(RND(S)*8):IF Q<>R THEN Q=R:GOTO 654+Q
651 IF X>T THEN 650

```



```

654 ? #J;"I discovered that ";TSUB$;" is not ";FACT$;".":GOTO Z
655 ? #J;"It was surprising to learn that ";TSUB$;" is not ";FACT$;".":GOTO Z
656 ? #J;"I wonder why being ";FACT$;" is not a quality of":? #J;TSUB$;". I'd li
ke to learn more.":GOTO Z
657 ? #J;"Will wonders never cease, ";TSUB$;" isn't ";FACT$;".":? #J;"Learning i
s such fun!":GOTO Z
658 ? #J;"However, what I found most interesting about ";TSUB$:? #J;"is that it'
s not ";FACT$;".":GOTO Z
659 ? #J;"You know, I kind of suspected all along that ";TSUB$;" wasn't":? #J;FA
CT$;". Now, I'm sure.":GOTO Z
660 ? #J;"I am truly enlightened! I finally learned that ";TSUB$;" is not":? #J;
FACT$;".":GOTO Z
661 ? #J;"As I learned more about ";TSUB$:? #J;"I also discovered that":? #J;TSU
B$;" is not ";FACT$;".
665 IF DEC$="Y" AND J=T THEN J=U:GOTO 654+Q
690 NEXT X:J=T:~
691 ? #J:IF E=T THEN ? #J;"Thanks...":? #J;"I have learned ";F(C)+N(C);" facts a
bout ";TSUB$;".":GOTO 730
692 IF CT<3 THEN 710
695 IF CT<7 THEN 720
700 ? #J;"Thanks for teaching me so much!":? #J;"My memory chips are dizzy!":GOT
O 730
710 ? #J;"I've already learned a little. Now, I am thirsty for more knowledge!":
GOTO 730
720 ? #J;"I feel like I have a good understanding about ";CT+T;" subjects.":? #J
;"Please teach me more."
730 IF DEC$="Y" AND J=T THEN J=U:GOTO 691
734 IF PEEK(53279)=5 THEN 734
735 POKE 752,S:GOTO D
740 DATA first things I discovered,things I now know,things I learned
741 DATA fact,thing,piece of information
742 DATA glad I learned,happy to discover,thrilled to learn
743 DATA very interesting,illuminating,enlightening
744 DATA enjoy,am pleased by,am saddened by
745 DATA admit that I was ignorant,confess my ignorance,admit how little I knew
746 DATA implications,a significance,a deeper meaning
747 DATA thank you for teaching me,emphasize the importance of knowing,thank you
for instructing me
779 REM ****TRANSITION TO NEW SUBJECT
780 IF W>S THEN 783
782 GOTO 820
783 X=X+T:K=810:IF W>6 THEN W=6
785 ? #J:GOTO 790+W
791 ? #J;" That's enough about ";QUE$;".":? #J;"Now I'd like to talk about ";TSU
B$;".":GOTO K
792 ? #J;" I don't feel like talking about":? #J;QUE$;" right now. I'd rather ta
lk about ";TSUB$;".":GOTO K
793 ? #J;" I find ";QUE$;" interesting and I may return to it,":? #J;"but let's
talk about ";TSUB$;" now.":GOTO K
794 ? #J;" I think I'll hold my thoughts about":? #J;QUE$;" momentarily and talk
about ";TSUB$;" instead.":GOTO K
795 ? #J;" Clearly, I'm very interested in":? #J;QUE$;" but I don't want to forg
et about ";TSUB$;" either!":GOTO K
796 ? #J;" I could go on and on about ";QUE$;" but that would be boring. Here's
what I learned about ";TSUB$;".
810 IF DEC$="Y" AND J=T THEN J=U:GOTO 785
820 W=S:J=T:RETURN

```



```

899 REM ****RELATE SUBJECT TO SUBJECT
900 C1=C:H=T:K=S:QUE$=FACT$:FOR C=S TO CT:CK=S:GOSUB 420:IF CK=S OR RND(S)<0.4 T
HEN 925
910 J=T:IF PEEK(53279)=5 THEN 910
915 IF TSUB$(C)SUB$(C*B+T,C*B+SUBL(C)) THEN I=INT(RND(S)*5):IF P<>I THEN P=I:?:X
=X+T:ON CK GOTO 960+P,970+P
920 IF TSUB$(C)SUB$(C*B+T,C*B+SUBL(C)) THEN 915
925 NEXT C:J=T
930 IF K>2 THEN ? #J;"Hmmm, several ";C$?:#J;"are not ";QUE$;".
940 IF H>5 THEN ? #J;"Wow! I just came up with ";H;" ";C$?:#J;"that are ";QUE$;
? #J;"like ";TSUB$;".!":GOTO 957
941 IF H>3 THEN ? #J;"Here's an interesting realization, many ";C$?:#J;"besides
";TSUB$;" are ";QUE$;".
957 IF DEC$="Y" AND J=T THEN J=U:GOTO 930
959 C=C1:RETURN
960 ? #J;"This is interesting since ";SUB$(C*B+T,C*B+SUBL(C));" is also ";QUE$;".
":GOTO G
961 ? #J;"In fact, ";TSUB$;" is ";QUE$;".":? #J;"and ";SUB$(C*B+T,C*B+SUBL(C));"
is too!":GOTO G
962 ? #J;"I have also learned that ";SUB$(C*B+T,C*B+SUBL(C));? #J;"is ";QUE$;" I
like ";TSUB$;".!":GOTO G
963 ? #J;"When I searched my memory, I remembered that":? #J;SUB$(C*B+T,C*B+SUBL
(C));" is ";QUE$;" too":GOTO G
964 ? #J;"After close scrutiny, I've discovered that ";SUB$(C*B+T,C*B+SUBL(C));?
#J;"is ";QUE$;" too,";
965 ? #J;" just like ";TSUB$;".":? #J;"So far, I have thought of ";H+T;" ";C$?:
#J;"that are ";QUE$;".":GOTO G
970 ? #J;"In contrast to ";TSUB$?:#J;SUB$(C*B+T,C*B+SUBL(C));" is not ";QUE$;"
at all":GOTO G
971 ? #J;"Of course, although ";TSUB$;" is ";QUE$?:#J;SUB$(C*B+T,C*B+SUBL(C));"
is not":GOTO G
972 ? #J;"I've also made a mental note that ";SUB$(C*B+T,C*B+SUBL(C));" is not "
;QUE$;" either":GOTO G
973 ? #J;"It's clear to me that although":? #J;TSUB$;" is ";QUE$;" ";SUB$(C*B+T,
C*B+SUBL(C));" is not":GOTO G
974 ? #J;"I thought about all ";CT+T;" ";C$?:#J;"you taught me, and discovered
that, unlike"
975 ? #J;TSUB$;". ";SUB$(C*B+T,C*B+SUBL(C));" is not ";QUE$;".":? #J;"I'm glad I
learned this now!"
978 ? :IF DEC$="Y" AND J=T THEN J=U:ON CK GOTO 960+P,970+P
980 IF CK=T THEN H=H+T
981 IF CK=U THEN K=K+T
998 W=W+T:GOTO 925
999 REM ****SAVE & RETRIEVE INFO
1000 TRAP 1120:?" )":? " Will you be using tape or disk":?(t/d)":INPUT X$:IF
X$="t" THEN 1025
1005 POKE 702,64:?"Enter filename D:":INPUT FACT$:QUE$="D:":QUE$(3)=FACT$:POKE
702,S:OPEN #3,8,S,QUE$:GOTO 1030
1025 ? " Cue tape. Press PLAY & RECORD and then RETURN when you hear two beep
s.":OPEN #3,8,S,"C:"
1030 POKE 559,S:?:#3;C$?:#3;F$?:#3;N$?:#3;SUB$:FOR X=S TO 9:FOR E=S TO 19:?:#
3;FL(X,E):NEXT E:NEXT X
1035 FOR X=S TO 9:FOR E=S TO 9:?:#3;FL1(X,E):NEXT E:NEXT X:FOR X=S TO 9:?:#3;SUB
L(X):NEXT X
1040 FOR X=S TO 9:?:#3;F(X):NEXT X:FOR X=S TO 9:?:#3;N(X):NEXT X:?:#3;CT:CLOSE #
3
1045 POKE 77,S:POKE 559,34:?" )":GOTO M

```



```

1050 TRAP 1120:?">":?" Will you be using tape or disk":?"(t/d)";:INPUT X$:IF
X$="t" THEN 1075
1055 POKE 702,64:?"Enter filename D:":;INPUT FACT$:QUE$="D:":QUE$(3)=FACT$:POKE
702,S:OPEN #3,4,S,QUE$:GOTO 1080
1075 ? " Cue tape. Press PLAY and then RETURN when you hear the beep.":OPEN #3,4
,S,"C:"
1080 POKE 559,S:FOR I=T TO 10:GET #3,P:C$(I,I)=CHR$(P):NEXT I:GET #3,P:FOR I=T
0 8000:GET #3,P:F$(I,I)=CHR$(P)
1085 NEXT I:GET #3,P:FOR I=T TO 4000:GET #3,P:N$(I,I)=CHR$(P):NEXT I:GET #3,P
1090 FOR I=T TO 200:GET #3,P:SUB$(I,I)=CHR$(P):NEXT I:GET #3,P
1095 FOR C=S TO 9:FOR E=S TO 19:INPUT #3,Y:FL(C,E)=Y:NEXT E:NEXT C
1100 FOR C=S TO 9:FOR E=S TO 9:INPUT #3,Y:FL1(C,E)=Y:NEXT E:NEXT C:FOR X=S TO 9:
INPUT #3,Y:SUBL(X)=Y:NEXT X
1105 FOR X=S TO 9:INPUT #3,Y:F(X)=Y:NEXT X:FOR X=S TO 9:INPUT #3,Y:N(X)=Y:NEXT X
:INPUT #3,CT
1110 CLOSE #3:C=CT:TSUB$=SUB$(C*B+T,C*B+SUBL(C)):POKE 77,S:POKE 559,34:?">":GOT
O M
1120 POKE 702,64:POKE 77,S:POKE 559,34:?"ERROR- ";PEEK(195);" AT LINE ";PEEK(
186)+256*PEEK(187):END
1199 REM ****READ IN DEMO INFORMATION
1200 POKE 752,T:?" Please be patient...":?" I am researching facts about a d
og,"?:?"and a panda."
1210 C$="Animals ":?" When I am done, I will have learned":?"8 facts about
what each animal is and"
1220 ? "4 facts about what each animal is not.I will then be delighted to tell y
ou all about them."
1230 RESTORE 1300:FOR CT=S TO T:READ TSUB$:SUBL(CT)=LEN(TSUB$):SUB$(CT*B+T,CT*B+
19)=TSUB$:NEXT CT
1240 CT=CT-T:FOR C=S TO CT:FOR I=T TO 8:READ FACT$:GOSUB 410:NEXT I:NEXT C
1250 FOR C=S TO CT:FOR I=T TO 4:READ FACT$:GOSUB 415:NEXT I:NEXT C:POKE 752,S:C=
C-T:RETURN
1300 DATA a panda,a dog
1310 DATA an endangered species,an inhabitant of western China,primarily a bambo
o eater
1311 DATA a member of the Order Carnivora,conspicuously black and white,about 3
lb when 6 weeks old
1312 DATA broad-skulled,more related to a raccoon than a bear
1320 DATA a member of the Family Canidae,available in many breeds,a member of th
e Order Carnivora
1321 DATA blind and deaf for 11 days after birth,Man's best friend,in heat twice
a year for 1-3 weeks
1322 DATA suseptable to many diseases,a common household pet
1330 DATA mature at less then 4 years-,a bear-,a common household pet-,found at
elevations less than 6000 ft-
1340 DATA broad-skulled-,an endangered species-,primarily a vegetarian-,a solita
ry hunter-

```